

```
1 ' -----
2 '       From the book Inside Active Directory, ISBN 0-201-61621-1
3 '       Copyright (C) 2002 by Addison-Wesley
4 '       Script by Sakari Kouti (see http://www.kouti.com)
5 ' You have a royalty-free right to use, modify, reproduce and distribute
6 ' this script (and/or any modified version) in any way you find useful,
7 ' provided that you agree that Addison-Wesley or Sakari Kouti has no
8 ' warranty, obligations or liability for the script. If you modify
9 ' the script, you must retain this copyright notice.
10 ' -----
11 Option Explicit
12
13 Const ADSTYPE_OCTET_STRING = 8
14 Const ADSTYPE_LARGE_INTEGER = 10
15 Const ADSTYPE_NT_SECURITY_DESCRIPTOR = 25
16
17 Dim objDSE, objADObject, i, strName, intGap
18 Dim objPropEntry, varPropValue, objPropValue, objLargeInt
19
20 Set objDSE = GetObject("LDAP://rootDSE")
21 Set objADObject = GetObject("LDAP://CN=Guest,CN=Users," & _
22     objDSE.Get("defaultNamingContext"))
23 objADObject.GetInfo
24
25 WScript.Echo "Number of properties in the property cache: " & _
26     objADObject.PropertyCount & vbCrLf
27
28 For i = 0 To objADObject.PropertyCount - 1
29     Set objPropEntry = objADObject.Item(i)
30     strName = objPropEntry.Name & " (" & objPropEntry.ADsType & ")"
31     For Each objPropValue In objPropEntry.Values
32         Select Case objPropEntry.ADsType
33             Case ADSTYPE_NT_SECURITY_DESCRIPTOR
34                 varPropValue = "some security descriptor"
35             Case ADSTYPE_OCTET_STRING
36                 varPropValue = "some octet string"
37             Case ADSTYPE_LARGE_INTEGER
38                 Set objLargeInt = _
39                     objPropValue.GetObjectProperty(objPropEntry.ADsType)
40                 varPropValue = _
41                     objLargeInt.HighPart * 2^32 + objLargeInt.LowPart
42             Case Else
43                 varPropValue = _
44                     objPropValue.GetObjectProperty(objPropEntry.ADsType)
45         End Select
46         intGap = 25 - Len(strName)
47         If intGap < 1 Then intGap = 1
48         WScript.Echo strName & String(intGap, " ") & varPropValue
49         strName = ""
50     Next
51 Next
52
```